Rules for integrands of the form $(d\,\mathrm{Sin}[e+f\,x])^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n$

1: $\int \mathrm{Sin}[e+f\,x]^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $\frac{m}{2}\in\mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis:** $\mathrm{Sin}[e+f\,x]^2 = \frac{\mathrm{Tan}[e+f\,x]^2}{1+\mathrm{Tan}[e+f\,x]^2}$
- **Basis: If** $\frac{m}{2}\in\mathbb{Z}$, **then** $\mathrm{Sin}[e+f\,x]^m\,F[b\,\mathrm{Tan}[e+f\,x]] = \frac{b}{f}\,\mathrm{Subst}\Big[\frac{x^m\,F[x]}{(b^2+x^2)^{\frac{m}{2}+1}},\,x,\,b\,\mathrm{Tan}[e+f\,x]\Big]\,\partial_x\,(b\,\mathrm{Tan}[e+f\,x])$
- **Rule: If** $\frac{m}{2}\in\mathbb{Z}$, **then**

$$\int \mathrm{Sin}[e+f\,x]^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx \longrightarrow \frac{b}{f}\,\mathrm{Subst}\Big[\int\frac{x^m\,(a+x)^n}{\left(b^2+x^2\right)^{\frac{m}{2}+1}}\,dx,\,x,\,b\,\mathrm{Tan}[e+f\,x]\Big]$$

- **Program code:**

```
Int[sin[e_.+f_.*x_]^m_*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b/f*Subst[Int[x^m*(a+x)^n/(b^2+x^2)^(m/2+1),x],x,b*Tan[e+f*x]] /;
FreeQ[{a,b,e,f,n},x] && IntegerQ[m/2]
```

2. $\int \mathrm{Sin}[e+f\,x]^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}$

1: $\int \mathrm{Sin}[e+f\,x]^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\,\bigwedge\,n\in\mathbb{Z}^+$

- **Derivation: Algebraic expansion**
- **Rule: If** $n\in\mathbb{Z}^+$, **then**

$$\int \mathrm{Sin}[e+f\,x]^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n\,dx \longrightarrow \int \mathrm{Expand}\big[\mathrm{Sin}[e+f\,x]^m\,(a+b\,\mathrm{Tan}[e+f\,x])^n,\,x\big]\,dx$$

- **Program code:**

```
Int[sin[e_.+f_.*x_]^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[Expand[Sin[e+f*x]^m*(a+b*Tan[e+f*x])^n,x],x] /;
FreeQ[{a,b,e,f},x] && IntegerQ[(m-1)/2] && IGtQ[n,0]
```

2: $\int \text{Sin}[e+f x]^m (a+b \text{Tan}[e+f x])^n \, dx$ when $\frac{m-1}{2} \in \mathbb{Z} \bigwedge n \in \mathbb{Z}^-$

- **Derivation: Algebraic expansion**
- **Basis:** $a+b \text{Tan}[z] == \frac{a \text{Cos}[z]+b \text{Sin}[z]}{\text{Cos}[z]}$

  Note: This rule sucks...
- **Rule: If** $\frac{m-1}{2} \in \mathbb{Z} \bigwedge n \in \mathbb{Z}^-$, **then**

$$\int \text{Sin}[e+f x]^m (a+b \text{Tan}[e+f x])^n \, dx \longrightarrow \int \frac{\text{Sin}[e+f x]^m (a \text{Cos}[e+f x]+b \text{Sin}[e+f x])^n}{\text{Cos}[e+f x]^n} \, dx$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[Sin[e+f*x]^m*(a*Cos[e+f*x]+b*Sin[e+f*x])^n/Cos[e+f*x]^n,x] /;
FreeQ[{a,b,e,f},x] && IntegerQ[(m-1)/2] && ILtQ[n,0] && (LtQ[m,5] && GtQ[n,-4] || EqQ[m,5] && EqQ[n,-1])
```

Rules for integrands of the form $(d \text{Csc}[e+f x])^m (a+b \text{Tan}[e+f x])^n$

1: $\int (d \text{Csc}[e+f x])^m (a+b \text{Tan}[e+f x])^n \, dx$ when $m \notin \mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \left((d \text{Csc}[e+f x])^m \left(\frac{\text{Sin}[e+f x]}{d}\right)^m\right) == 0$
- **Rule: If** $m \notin \mathbb{Z}$, **then**

$$\int (d \text{Csc}[e+f x])^m (a+b \text{Tan}[e+f x])^n \, dx \longrightarrow (d \text{Csc}[e+f x])^{\text{FracPart}[m]} \left(\frac{\text{Sin}[e+f x]}{d}\right)^{\text{FracPart}[m]} \int \frac{(a+b \text{Tan}[e+f x])^n}{\left(\frac{\text{Sin}[e+f x]}{d}\right)^m} \, dx$$

- **Program code:**

```
Int[(d_.*csc[e_.+f_.*x_])^m_*(a_.+b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  (d*Csc[e+f*x])^FracPart[m]*(Sin[e+f*x]/d)^FracPart[m]*Int[(a+b*Tan[e+f*x])^n/(Sin[e+f*x]/d)^m,x] /;
FreeQ[{a,b,d,e,f,m,n},x] && Not[IntegerQ[m]]
```

Rules for integrands of the form $\text{Cos}[e+f x]^m \text{Sin}[e+f x]^p (a+b \text{Tan}[e+f x])^n$

1: $\int \text{Cos}[e+f x]^m \text{Sin}[e+f x]^p (a+b \text{Tan}[e+f x])^n \, dx$ when $n \in \mathbb{Z}$

- **Derivation: Algebraic simplification**
- **Basis:** $a+b \text{Tan}[z] == \frac{a \text{Cos}[z]+b \text{Sin}[z]}{\text{Cos}[z]}$
- **Rule: If** $n \in \mathbb{Z}$**, then**

$$\int \text{Cos}[e+f x]^m \text{Sin}[e+f x]^p (a+b \text{Tan}[e+f x])^n \, dx \longrightarrow \int \text{Cos}[e+f x]^{m-n} \text{Sin}[e+f x]^p (a \text{Cos}[e+f x]+b \text{Sin}[e+f x])^n \, dx$$

- **Program code:**

```
Int[cos[e_.+f_.*x_]^m_.*sin[e_.+f_.*x_]^p_.*(a_+b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[Cos[e+f*x]^(m-n)*Sin[e+f*x]^p*(a*Cos[e+f*x]+b*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,m,p},x] && IntegerQ[n]
```

```
Int[sin[e_.+f_.*x_]^m_.*cos[e_.+f_.*x_]^p_.*(a_+b_.*cot[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[Sin[e+f*x]^(m-n)*Cos[e+f*x]^p*(a*Sin[e+f*x]+b*Cos[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,m,p},x] && IntegerQ[n]
```